# TP Maple 2 | Suites et séries de fonctions. Séries entières.

*On utilisera avec profit les outils graphiques du logiciel afin d'expérimenter et d'émettre des conjectures. La manipulation des séries entières se fera au moyen des librairies* **powseries** *et* **gfun** *de Maple.*



## 1. Suites de fonctions

Illustrons pour commencer la convergence simple de la suite $(f_n)_{n\geqslant 1}$ définie par

$$f_n : x \longmapsto f_n(x) = \begin{cases} \left(1-\dfrac{x}{n}\right)^n & \text{si } 0 \leqslant x \leqslant n \\ 0 & \text{si } x > n \end{cases} \qquad \text{vers } x \longmapsto e^{-x} \text{ sur } \mathbb{R}_+$$

```
> f:=piecewise(x<n,(1-x/n)^n,0): plot([seq(subs(n=k,f),k=1..8),exp(-x)],
  x=0..5, color=[seq(blue,k=1..8),red], thickness=[seq(1,k=1..8),3]);
```



On pourra conjecturer le mode de convergence d'une suite de fonctions $(f_n)_{n\geqslant 0}$ donnée en faisant une étude graphique au moyen de la commande **plot**. Une animation permettra de mieux apréhender le mode convergence de la suite $(f_n)_{n\geqslant 0}$. On pourra vérifier les résultats en utilisant des commandes telles que **limit**, en calculant $\|f_n\|_\infty$ (par une étude de fonction), etc.

Etudions à présent la suite définie par

$$\forall n \in \mathbb{N},\ \ \forall x \in [0,1],\ \ f_n(x) = n \cdot x^n \cdot (1-x)$$

On commence par tracer sur un même graphique les courbes représentatives des fonctions $f_0, f_1, \ldots, f_{20}$.

```
> f:=n*x^n*(1-x): plot([seq(subs(n=k,f),k=0..20)],x=0..1,color=blue);
```




On conjecture qu'il y a convergence simple vers la fonction nulle mais que cette convergence n'est pas uniforme car la suite des maximums semble converger vers une limite non nulle. Voyons cela de plus près...

```
> sol:=solve({diff(f,x)},x);
```

$$sol := \left\{\frac{n}{n+1}\right\}$$

```
> u:=subs(sol,f): limit(u,n=infinity);
```

$$e^{-1}$$

Rappelons la syntaxe permettant une animation sous Maple :

La commande animate

**animate(plotcommand, plotargs, t=a..b, options)**

On pourra essayer la ligne suivante...

```
> with(plots):
  animate(plot,[n*x^n*(1-x),x=0..1], n=0..50);
```

**Exercice 1.**

Pour tout entier naturel $n$ et tout réel $x$, on pose

$$f_n(x) = \sin(x) \cdot \cos^n(x)$$

1. Conjectuer le mode de convergence de $(f_n)_{n\geqslant 0}$ sur $\mathbb{R}$ au moyen d'une animation sous Maple.
2. Démontrer ce résultat en vous aidant du logiciel pour les calculs intérmédiaires.

**Exercice 2.**

Soit $\alpha > 0$. Etudier au moyen de Maple le mode de convergence la suite de fonctions définie par

$$\begin{array}{rccl} f_n : & [0,1] & \longrightarrow & \mathbb{R} \\ & x & \longmapsto & n^\alpha \cdot x^n \cdot \sin(\pi x) \end{array}$$

## 2. Séries de fonctions

On utilisera la commande **sum** pour définir la somme d'une série de fonction. On n'hésitera pas également à étudier les sommes partielles pour se faire une idée.

Posons-nous la question du mode de convergence sur $\mathbb{R}_+$ de la série :

$$f(x) = \sum_{n=1}^{+\infty} (-1)^n \cdot \ell\text{n}\left(1 + \frac{x}{n\cdot(x+1)}\right)$$

On commence par vérifier que $f$ est définie sur $\mathbb{R}_+$ :

```
> u:=:=(-1)^(n)*ln(1+x/(n*(1+x))):
> series(u,n =infinity,2);
```

$$\frac{(-1)^n x}{(x+1)n} + O\left(\frac{1}{n^2}\right)$$

On peut aussi remarquer que cette série vérifie les hypothèses du critère spécial des séries alternées. Voici quelques restes...

```
> restes:=sum(u,n=N..infinity):
  plot([seq(subs(N=5*m,restes),m=1..10)],x=0..infinity,color=blue);
```




Il semble que la convergence soit uniforme sur $\mathbb{R}_+$. Démontrons-le. On a classiquement

$$\forall x \in \mathbb{R}_+,\quad \left|\sum_{n=N}^{+\infty}(-1)^n\cdot\ell\mathrm{n}\left(1+\frac{x}{n\cdot(x+1)}\right)\right| \leqslant \ell\mathrm{n}\left(1+\frac{x}{N\cdot(x+1)}\right) \leqslant \frac{1}{N}\cdot\frac{x}{x+1} \leqslant \frac{1}{N}$$

Il y a donc convergence uniforme sur $\mathbb{R}_+$. La convergence n'est clairement pas normale sur $\mathbb{R}_+$.

**Exercice 3.**

Soit, pour tout entier narurel $n$ non nul,

$$f_n : x \mapsto x\cdot e^{-n^2\cdot x^2}$$

1. Représenter simultanément $f_1$, $f_2$ et $f_3$.
2. Etablir la convergence de la série $\sum f_n$ sur $\mathbb{R}_+$. On note $f$ sa somme.
3. Représenter les premières sommes partielles de $f$. Représenter $f$.
4. *(Réservé aux 5/2)* La fonction $f$ est-elle bornée ?

**Exercice 4.**

On pose,

$$u_n(x) = \frac{1}{n + x \cdot n^2} \quad \text{et} \quad S(x) = \sum_{n=1}^{+\infty} u_n(x)$$

**1**. Déterminer l'ensemble de définition de $S$.

**2**. En utilisant Maple, tracer le graphe de $S$.

**3**. Déterminer un équivalent de $S(x)$ lorsque $x$ tend vers $+\infty$ en utilisant le logiciel. Prouver le résultat obtenu.

### 3. Cas particulier des séries entières

Les sommes classiques pourront être calculée au moyen de la commande **sum**.

```
> sum(x^n/n,n=1..infinity);
```

$$-\ell\text{n}(1-x)$$

Tout autre manipulation sur les séries entières pourra se fera en chargeant au préalable une librairie ad hoc. Il existe deux librairies sous Maple permettant la manipulation formelle des séries entières. Nous en donnerons un bref aperçu dans les deux paragraphes qui suivent.

#### 3.1. La librairie powseries

En utilisant la librairie **powseries** [1] de Maple, on pourra manipuler les séries entières.

```
> with(powseries);
```

[compose, evalpow, inverse, multconst, multiply, negative, powadd, powcos, powcreate, powdiff, powexp, powint, powlog, powpoly, powsin, powsolve, powsqrt, quotient, reversion, subtract, tpsform]

On pourra créer la série entière définie par une suite $(u_n)_{n\geqslant 0}$ au moyen de la commande **powcreate**. On remarquera que cette commande accepte les relations de récurrences [2].

1. *Séries entières* se dit *power series* en anglais.
2. Mais pas toutes...

```
> powcreate(u(n)=1): powcreate(v(n)=v(n-1)+1,v(0)=1)): w:=multiply(u,v):
  tpsform(w, x, 10);
```

$$1+3x+6x^2+10x^3+15x^4+21x^5+28x^6+36x^7+45x^8+55x^9+O(x^{10})$$

La série entière créée par **powcreate(u(n)=...)** sera automatiquement appelée $u$. La fonction **tpsform** permet d'écrire la troncature[3] d'une série entière donnée à un ordre arbitrairement choisi. Il ne s'agit ni plus ni moins que du développement de Taylor en 0 de la somme de cette série. On pourra accéder à la valeur du coefficient $w_n$ par **w(n)** mais le logiciel ne retournera le résultat que pour des valeurs numériques de $n$.

```
> w(2), w(0), w(n);
```

$$6, 1, w(n)$$

Les séries entières définies précédemment sont bien connues des lecteurs et vérifient clairement

$$\forall x \in ]-1,1[,\quad u(x)=\sum_{n=0}^{+\infty} x^n = \frac{1}{1-x},\quad v(x)=u'(x)=\sum_{n=1}^{+\infty} n x^{n-1} = \frac{1}{(1-x)^2}$$

$$\text{et}\quad w(x)=u(x)\cdot v(x)=u(x)\cdot u'(x)=\frac{1}{(1-x)^3}=\frac{(u^2)'(x)}{2}.$$

On peut vérifier cette dernière égalité sur les troncatures mais pas davantage...

```
> f:=evalpow(u^2): g:=powdiff(f): h:=multconst(g,1/2): tpsform(h,x,10);
```

$$1+\frac{x^2}{2}+\frac{x^4}{8}+\frac{x^6}{48}+\frac{x^8}{384}+O(x^{10})$$

La commande **powsolve** permet de trouver les solutions développables en séries entières d'une équation différentielle donnée (avec ou sans conditions initiales).

```
> f:=powsolve({diff(y(x),x)-y(x)=0, y(0)=1}): tpsform(f, x, 4);
```

$$1+x+\frac{x^2}{2}+\frac{x^3}{6}+O(x^4)$$

Chacun aura reconnu la fonction exponentielle.

3. *Truncated power series* in english.

Séries entières avec la librairie **powseries**

- **powcreate(*expression de* $u_n$)** : crée la série entière $u$ définie par $\sum u_n \cdot x^n$.
- **powcreate(*relation de récurrence sur* $u_n$ *avec conditions initiales*)** : crée la série entière $u$ définie par $\sum u_n \cdot x^n$.
- **f :=powsolve(*EDL avec conditions initiales*)** : en cas d'existence, crée la solution développable en série entière de l'EDL donnée.
- **tpsform(f,x,n)** : renvoie le développement de Taylor de $f$ en 0 (la variable étant $x$) à l'ordre $n$.
- **u(k)** : renvoie la valeur de $u_k$ pour $u = \sum u_n \cdot x^n$.

**Exercice 5.**

Soit $f : \mathbb{R} \to \mathbb{R}$ vérifiant

$$\forall x \in \mathbb{R},\ 2 \cdot f(x)^2 - f(2 \cdot x) = 1$$

On suppose que $f$ est développable en série entière au voisinage de 0. On écrit alors, au voisinage de 0,

$$f(x) = \sum_{n=0}^{+\infty} \frac{u_n}{n!} \cdot x^n$$

En utilisant le logiciel, conjecturer les expressions possibles de $f(x)$.

**Exercice 6.**

Soit $(u_n)_{n\geqslant 0}$ vérifiant $\forall n \geqslant 0$,

$$u_{n+3} = u_{n+2} + u_{n+1} - u_n$$

On admet ici que le rayon de convergence de la série entière

$$\sum_{n\geqslant 0} u_n \cdot x^n$$

est non nul. On note $S$ sa somme.

**1.** Calculer explicitement $S(x)$ en utilisant Maple. On pourra utiliser la commande **convert** avec l'option **ratpoly**.

**2.** En déduire l'expression de $u_n$ en fonction de $n$.

On l'aura compris au travers de ces quelques exemples, la librairie **powseries** sera d'un intérêt limité dans les cas où l'on recherche des développement en série entière car aucune commande ne permet d'accéder à l'expression générale du coefficient $u_n$. La librairie **gfun** sera plus utile [4] dans ce contexte.

4. mais n'attendez pas trop de miracles tout de même...

### 3.2. La librairie gfun

Commençons par quelques définitions sur les séries génératrices. A toute suite de nombre complexe $(a_n)_{n\geqslant 0}$ on peut associer deux séries :

$$\sum_{n\geqslant} a_n \cdot z^n \quad \text{et} \quad \sum_{n\geqslant 0} \frac{a_n}{n!} \cdot z^n$$

appellées respectivement *série génératrice (ordinaire)* et *série génératrice exponentielle* de la suite $(a_n)_{n\geqslant}$. Sous réserve de convergence, ces séries permettent souvent d'obtenir des renseignements utiles [5] sur la suite $(a_n)_{n\geqslant 0}$. Citons par exemple la résolution des relations de récurrence, i.e. le calcul explicite de $a_n$ en fonction de $n$ lorsque $(a_n)_{n\geqslant 0}$ vérifie une relation de récurrence donnée. En utilisant la librairie **gfun** [6] de Maple, on pourra manipuler les séries génératrices.

```
> with(gfun);
```

[Laplace, Parameters, algebraicsubs, algeqtodiffeq, algeqtoseries, algfuntoalgeq, borel, cauchyproduct, diffeq*diffeq, diffeq+diffeq, diffeqtohomdiffeq, diffeqtorec, gfunParameters, guesseqn, guessgf, hadamardproduct, holexprtodiffeq, invborel, listtoalgeq, listtodiffeq, listtohypergeom, listtolist, listtoratpoly, listtorec, listtoseries, poltodiffeq, poltorec, ratpolytocoeff, rec*rec, rec+rec, rectodiffeq, rectohomrec, rectoproc, seriestoalgeq, seriestodiffeq, seriestohypergeom, seriestolist, seriestoratpoly, seriestorec, seriestoseries]

Sans entrer dans les détails, l'utilisateur de **gfun** pourra notament déduire d'une EDL **(E)** la relation de récurrence vérifiée par les coefficients d'une solution de **(E)** développable en série entière. Les outils de résolution de Maple et en particulier **rsolve**, le solveur de récurrences du logiciel, pourront s'avérer utiles dans ce contexte pour expliciter les coefficients. Revenons à l'exemple du paragraphe précédent.

```
> rec:=diffeqtorec({(1-z)*(diff(y(z),z))-3*y(z)=0,y(0)=1},y(z),v(n)),
  rsolve(rec,v);
```

$$v(0)=1, (-3-n)\,v(n)+(n+1)\,v(n+1), \frac{1}{2}(n+2)(n+1)$$

On en déduit donc que

$$\forall x \in ]-1,1[, \quad \frac{1}{(1-x)^3} = \sum_{n=0}^{+\infty} \frac{(n+1)\cdot(n+2)}{2} \cdot x^n$$

5. Elles sont également très utilisées pour le calcul des probabilités.
6. *Fonction génératrice* se dit *generating function* en anglais.

La librairie **gfun**

► **diffeqtorec** : permet de trouver la relation de récurrence linéaire à coefficients polynoniaux correspondant à une équation différentielle linéaire à coefficients polynomiaux.

► **rectodiffeq** : permet de trouver l'équation différentielle linéaire à coefficients polynomiaux correspondant à une relation de récurrence linéaire à coefficients polynomiaux.

**Exercice 7.**

Soient $k \in \mathbb{R}$ et $(a_n)_{n \geqslant 0}$ la suite définie par

$$a_0 = 1, \ \ \forall n \geqslant 0, \ \ a_{n+1} = \frac{2 \cdot n + k}{n+1} \cdot a_n$$

On note

$$f(x) = \sum_{n=0}^{+\infty} a_n x^n$$

**1**. Montrer que le rayon de convergence de cette série est non nul.

**2**. En utilisant Maple, trouver une équation différentielle vérifiée par $f$ au voisinage de 0.

**3**. Toujours à l'aide du logiciel, expliciter $f(x)$.